Introduction to xCoba

Downloading and Installing xAct

Visit the page: http://www.xact.es
Follow installation instructions on http://www.xact.es/download.html
See also my “Introduction to xTensor” video: https://youtu.be/pQwrKF08Xps

Linux:
1. Download the tarball xAct_V.tgz (V is the version number)
2. sudo -i ; cd /usr/share/Mathematica/Applications/; tar xvfz ~/Downloads/xAct_V.tgz

Windows:
1. Download the zip file xAct_V.zip (V is the version number)
2. unzip its contents in C:\Program Files\Wolfram Research\Mathematica\<version>\AddOns\Applications\

Read the documentation:
http://www.xact.es/documentation.html
If you want to use xCoba, you will not avoid reading the full documentation. Better earlier than later: xCobaDoc.nb

The documentation is also installed locally, most likely in:
Linux: /usr/share/Mathematica/Applications/xAct/Documentation/English/
Windows: C:\Program Files\Wolfram Research\Mathematica\<version>\AddOns\Applications\xAct\Documentation\English

Explore the documentation in xCobaDoc.nb.... Make a copy to the notebook, so that you can play with it.

```
cp /usr/share/Mathematica/Applications/xAct/Documentation/English/xCobaDoc.nb .
```

Start an xCoba session

If you are already running a Mathematica session, esp. if you have loaded xTensor, xAct, ..., make a call to Quit[] before the loading of the package:
Quit[]; Needs[“xAct`xCoba`”]

```
In[◦]:= Needs["xAct`xCoba`"]
```

```
------------------------------------------------------------
Package xAct`xPerm`  version 1.2.3, {2015, 8, 23}
CopyRight (C) 2003-2018, Jose M. Martin-Garcia, under the General Public License.
Connecting to external linux executable...
Connection established.
------------------------------------------------------------
Package xAct`xTensor`  version 1.1.3, {2018, 2, 28}
CopyRight (C) 2002-2018, Jose M. Martin-Garcia, under the General Public License.
------------------------------------------------------------
Package xAct`xCoba`  version 0.8.4, {2018, 2, 28}
CopyRight (C) 2005-2018, David Yllanes and
  Jose M. Martin-Garcia, under the General Public License.
------------------------------------------------------------
These packages come with ABSOLUTELY NO WARRANTY; for details type
  Disclaimer[]. This is free software, and you are welcome to redistribute
  it under certain conditions. See the General Public License for details.
------------------------------------------------------------
```

Define a manifold: you need a name, its dimension, and the set of indices that you will use:
Same as in xTensor: it also defines the associated Tangent Bundle Tangent<Name>
In our case: M and TangentM

```
In[◦]:= DefManifold[M, 4, {λ, μ, ν, ρ, σ, α, β, γ, δ}];
** DefManifold: Defining manifold M.
** DefVBundle: Defining vbundle TangentM.
```

```
In[◦]:= dimM = DimOfManifold[M]; dimM1 = dimM - 1;
```

Definition of a chart

For xCoba, a chart is synonymous to a coordinate basis.

A set of scalar fields (in the sense of xTensor) is defined with it, the coordinates.

The basis of the chart gets its color, this distinguishes it in the output, when there are more bases present, from abstract indices etc.

Non coordinate bases can also be defined - not presented here -, the main difference being that their “partial derivative” does not commute on functions (there is torsion, their Lie bracket is non vanishing etc). The terminology for “partial derivative” in xCoba is “Parallel Derivative”.

In the definition below:
ch: the name of the chart (or coordinate basis)
{0,1,2,3}: the allowed values of indices. There is no restriction except they have to be as many as dimM
{t[],r[],θ[],ϕ[]}: the names of the scalars that will be used for coordinates. Note the (compulsory) square brackets.
ChartColor: the color used for the indices of the basis ch. If omitted, a default color is used

In[ ]:=
```
DefChart[ch, M, {0, 1, 2, 3}, {t[], r[], θ[], ϕ[]}, ChartColor → Blue]
```

```
** DefChart: Defining chart ch.
** DefTensor: Defining coordinate scalar t[].
** DefTensor: Defining coordinate scalar r[].
** DefTensor: Defining coordinate scalar θ[].
** DefTensor: Defining coordinate scalar ϕ[].
** DefMapping: Defining mapping ch.
** DefMapping: Defining inverse mapping ich.
** DefTensor: Defining mapping differential tensor dich[-a, ichα].
** DefTensor: Defining mapping differential tensor dch[-α, cha].
** DefBasis: Defining basis ch. Coordinated basis.
** DefCovD: Defining parallel derivative PDch[-α].
** DefTensor: Defining vanishing torsion tensor TorsionPDch[α, -β, -γ].
** DefTensor: Defining symmetric Christoffel tensor ChristoffelPDch[α, -β, -γ].
** DefTensor: Defining vanishing Riemann tensor RiemannPDch[-α, -β, -γ, δ].
** DefTensor: Defining vanishing Ricci tensor RicciPDch[-α, -β].
** DefTensor: Defining antisymmetric +1 density etaUpch[α, β, γ, δ].
** DefTensor: Defining antisymmetric -1 density etaDownch[-α, -β, -γ, -δ].
```

We can use tensor objects, as in xTensor. The coordinates have been defined as scalars t[ ], r[ ], θ[ ], ϕ[ ], and can be used in any expression.

xTensor expressions will be treated as abstract, non coordinated objects:

In[ ]:=
```
DefConstantSymbol[mass   , PrintAs → "M"];
DefScalarFunction[scalar, PrintAs → "Φ"];
DefScalarFunction[s ];
DefTensor[w[μ], M];
{mass, mass scalar[], 2 mass/r[], s[] scalar[] r[] w[μ]}
```

```
** DefConstantSymbol: Defining constant symbol mass.
** DefScalarFunction: Defining scalar function scalar.
** DefScalarFunction: Defining scalar function s.
** DefTensor: Defining tensor w[μ].
```

Out[ ]= $\left\{M, M\,\Phi[], \frac{2\,M}{r}, r\,s[]\,\Phi[]\,w^{\mu}\right\}$

The basic object used as a tensor with components, is a CTensor object, defined by the CTensor function.
After that, the object can be used as an abstract tensor, or perform contractions w.r.t. chosen indices.

In the following example, v is a vector with:
{2, -1, 3, 4}: its components
{ch}: the basis w.r.t. the components are calculated. For a 1-form, we would have used {-ch}.

In[ ]:=
```
v = CTensor[{2, -1, 3, 4}, {ch}]
```

Out[ ]=
```
CTensor[{2, -1, 3, 4}, {ch}, 0]
```

StandardForm of the vector. It is printed with one abstract index.
This is as an xTensor object, except that the head displays the components.
That the components refer to the ch-basis is indicated by the blue color (the basis' color)

In[ ]:=
```
v[μ]
```
Out[ ]=

$$\left|\begin{matrix}2\\-1\\3\\4\end{matrix}\right|^{\mu}$$

In[ ]:=
```
{StandardForm[v[μ]], TraditionalForm[v[μ]],
 OutputForm[v[μ]], InputForm[v[μ]], FullForm[v[μ]]}
```
Out[ ]=

$\left\{\left|\begin{matrix}2\\-1\\3\\4\end{matrix}\right|^{\mu}\right.$, CTensor({2, -1, 3, 4}, {ch}, 0)(μ), CTensor[{2, -1, 3, 4}, {ch}, 0][μ],

CTensor[{2, -1, 3, 4}, {ch}, 0][μ], CTensor[List[2, -1, 3, 4], List[ch], 0][\[Mu]]}

A CTensor can take 3 types of indices:

Type		Example	
Abstract	index:	μ	
Basis	index:	{μ,ch}	ch is the name of the chart the index refers to
Component	index:	{0,ch}	the number must be in the list of allowed index values {0,1,2,3} defined for the chart ch

The following expands v^{μ} as a liner combination of the basis vectors of the chart ch. The blue numbers indicate that they are the (coordinate) basis vectors of ch. The index μ is abstract (abstract index notation).

In[ ]:=
```
v[{μ, ch}]
```
Out[ ]=

$$2\, e_0{}^{\mu} - e_1{}^{\mu} + 3\, e_2{}^{\mu} + 4\, e_3{}^{\mu}$$

v[{α,ch}], α=0,1,2,3 gives the respective components:

In[ ]:=
```
{v[{0, ch}], v[{1, ch}], v[{2, ch}], v[{3, ch}]}
```
Out[ ]=
```
{2, -1, 3, 4}
```

Example of a one form:

In[ ]:=
```
ω = CTensor[{-2, 1, -4, 3}, {-ch}]
```
Out[ ]=
```
CTensor[{-2, 1, -4, 3}, {-ch}, 0]
```

Output form of the 1-form. It is printed with one abstract index.
This is as an xTensor object, except that the head displays the components.
That the components refer to the ch-basis is indicated by the blue color (the basis' color)

In[ ]:=
```
ω[-μ]
```
Out[ ]=
$$\begin{vmatrix} -2 \\ 1 \\ -4 \\ 3 \end{vmatrix}_{\mu}$$

Here, we see the dual basis used in the expansion:

In[ ]:=
```
ω[{-μ, -ch}]
```
Out[ ]=
$$-2\, e_{\mu}{}^{0} + e_{\mu}{}^{1} - 4\, e_{\mu}{}^{2} + 3\, e_{\mu}{}^{3}$$

Notice, there is no - (minus sign) when a component index is used. The (-) must be put in front of the basis:

In[ ]:=
```
Table[ω[{α, -ch}], {α, 0, 3}]
```
Out[ ]=
```
{-2, 1, -4, 3}
```

If omitted, xCoba does not know what to do. This will change when we will define a metric.

In[ ]:=
```
Table[ω[{α, ch}], {α, 0, 3}]
```
MetricsOfVBundle: There is no metric in TM.

Throw: Uncaught Throw[Null] returned to top level.

Out[ ]=
```
Hold[Throw[Null]]
```

We can do contractions:

In[ ]:=
```
ω[-μ] v[μ]
```
Out[ ]=
```
-5
```

We can contract with an abstract tensor. The output form of the contraction is the same as in xTensor, except for the head of the CTensor object that displays the components.

In[ ]:=
```
ω[-μ] w[μ]
```

Out[ ]=
$$w^{\mu}\ \left|\begin{matrix}-2\\1\\-4\\3\end{matrix}\right|_{\mu}$$

As a linear combination of the basis covectors

In[ ]:=
```
ω[-μ] w[μ] // ToBasisExpand
```

Out[ ]=
$$\left(-2\, e_{\mu}{}^{0}+e_{\mu}{}^{1}-4\, e_{\mu}{}^{2}+3\, e_{\mu}{}^{3}\right) w^{\mu}$$

An an expression using components of w^{μ} in the ch-basis. The color of the indices indicate the basis that the components correspond to.

In[ ]:=
```
ω[-μ] w[μ] // ToBasisExpand  // ContractBasis
```

Out[ ]=
$$-2\, w^{0}+w^{1}-4\, w^{2}+3\, w^{3}$$

We can do ordinary tensor algebra, using the xTensor notation:

In[ ]:=
```
{3 v[μ], 7 v[μ] + 2 w[μ], v[μ] v[ν], v[μ] w[ν], ω[-μ] v[ν]}
```

Out[ ]=
$$\left\{\left|\begin{matrix}6\\-3\\9\\12\end{matrix}\right|_{\mu},\ 2\, w^{\mu}+\left|\begin{matrix}14\\-7\\21\\28\end{matrix}\right|_{\mu},\ \boxed{\begin{matrix}4&-2&6&8\\-2&1&-3&-4\\6&-3&9&12\\8&-4&12&16\end{matrix}}_{\mu\nu},\ w^{\nu}\left|\begin{matrix}2\\-1\\3\\4\end{matrix}\right|_{\mu},\ \boxed{\begin{matrix}-4&2&-6&-8\\2&-1&3&4\\-8&4&-12&-16\\6&-3&9&12\end{matrix}}_{\mu}{}^{\nu}\right\}$$

In[ ]:=
```
{3 v[μ], 7 v[μ] + 2 w[μ], v[μ] v[ν], v[μ] w[ν], ω[-μ] v[ν]} // ToBasisExpand
```

Out[ ]=
$$\{6\, e_0{}^{\mu}-3\, e_1{}^{\mu}+9\, e_2{}^{\mu}+12\, e_3{}^{\mu},\ 14\, e_0{}^{\mu}-7\, e_1{}^{\mu}+21\, e_2{}^{\mu}+28\, e_3{}^{\mu}+2\, w^{\mu},$$
$$4\, e_0{}^{\mu}\ e_0{}^{\nu}-2\, e_0{}^{\nu}\ e_1{}^{\mu}-2\, e_0{}^{\mu}\ e_1{}^{\nu}+e_1{}^{\mu}\ e_1{}^{\nu}+6\, e_0{}^{\nu}\ e_2{}^{\mu}-$$
$$3\, e_1{}^{\nu}\ e_2{}^{\mu}+6\, e_0{}^{\mu}\ e_2{}^{\nu}-3\, e_1{}^{\mu}\ e_2{}^{\nu}+9\, e_2{}^{\mu}\ e_2{}^{\nu}+8\, e_0{}^{\nu}\ e_3{}^{\mu}-4\, e_1{}^{\nu}\ e_3{}^{\mu}+$$
$$12\, e_2{}^{\nu}\ e_3{}^{\mu}+8\, e_0{}^{\mu}\ e_3{}^{\nu}-4\, e_1{}^{\mu}\ e_3{}^{\nu}+12\, e_2{}^{\mu}\ e_3{}^{\nu}+16\, e_3{}^{\mu}\ e_3{}^{\nu},$$
$$\left(2\, e_0{}^{\mu}-e_1{}^{\mu}+3\, e_2{}^{\mu}+4\, e_3{}^{\mu}\right) w^{\nu},\ -4\, e_{\mu}{}^{0}\ e_0{}^{\nu}+2\, e_{\mu}{}^{1}\ e_0{}^{\nu}-8\, e_{\mu}{}^{2}\ e_0{}^{\nu}+6\, e_{\mu}{}^{3}\ e_0{}^{\nu}+$$
$$2\, e_{\mu}{}^{0}\ e_1{}^{\nu}-e_{\mu}{}^{1}\ e_1{}^{\nu}+4\, e_{\mu}{}^{2}\ e_1{}^{\nu}-3\, e_{\mu}{}^{3}\ e_1{}^{\nu}-6\, e_{\mu}{}^{0}\ e_2{}^{\nu}+3\, e_{\mu}{}^{1}\ e_2{}^{\nu}-$$
$$12\, e_{\mu}{}^{2}\ e_2{}^{\nu}+9\, e_{\mu}{}^{3}\ e_2{}^{\nu}-8\, e_{\mu}{}^{0}\ e_3{}^{\nu}+4\, e_{\mu}{}^{1}\ e_3{}^{\nu}-16\, e_{\mu}{}^{2}\ e_3{}^{\nu}+12\, e_{\mu}{}^{3}\ e_3{}^{\nu}\}$$

In[ ]:=
```
v[μ] w[ν] // ToBasisExpand // ContractBasis
```

Out[ ]=
$$2\, e_0{}^{\mu}\ w^{\nu}-e_1{}^{\mu}\ w^{\nu}+3\, e_2{}^{\mu}\ w^{\nu}+4\, e_3{}^{\mu}\ w^{\nu}$$

In[ ]:=
```
ω[-μ] w[ν] // ToBasisExpand // ContractBasis
```
Out[ ]=
$$-2\, e_\mu{}^{0}\; w^\nu + e_\mu{}^{1}\; w^\nu - 4\, e_\mu{}^{2}\; w^\nu + 3\, e_\mu{}^{3}\; w^\nu$$

We can assign CTensors with “=”

In[ ]:=
$$u1 = \frac{2\,mass}{r[]}\, v$$
Out[ ]=
$$\text{CTensor}\left[\left\{\frac{4\,M}{r}, -\frac{2\,M}{r}, \frac{6\,M}{r}, \frac{8\,M}{r}\right\}, \{ch\}, 0\right]$$

In[ ]:=
```
ω[-μ] u1[μ]
```
Out[ ]=
$$-\frac{10\,M}{r}$$

We have to convert w to a CTensor object to make the assignment.

In[ ]:=
```
ToCTensor[w, {ch}]
```
Out[ ]=
$$\text{CTensor}[\{w^{0}, w^{1}, w^{2}, w^{3}\}, \{ch\}, 0]$$

w remains an xTensor abstract vector

In[ ]:=
```
w[μ]
```
Out[ ]=
$$w^{\mu}$$

But we can use ToCTensor to construct more complicated CTensor expressions

In[ ]:=
$$u = \left(1 - \frac{2\,mass}{r[]}\right) \text{ToCTensor}[w, \{ch\}]$$
Out[ ]=
$$\text{CTensor}\left[\left\{\left(1-\frac{2\,M}{r}\right) w^{0}, \left(1-\frac{2\,M}{r}\right) w^{1}, \left(1-\frac{2\,M}{r}\right) w^{2}, \left(1-\frac{2\,M}{r}\right) w^{3}\right\}, \{ch\}, 0\right]$$

In[ ]:=
```
ω[-μ] u[μ]
```
Out[ ]=
$$-2\left(1-\frac{2\,M}{r}\right) w^{0} + \left(1-\frac{2\,M}{r}\right) w^{1} - 4\left(1-\frac{2\,M}{r}\right) w^{2} + 3\left(1-\frac{2\,M}{r}\right) w^{3}$$

More complicated expressions. Notice the [ ] when a scalar is used.

In[ ]:=
```
scalar[] ToCTensor[w, {ch}] / (r[]^2 Sin[θ[]]^2)
```

Out[ ]=

$$\text{CTensor}\left[\left\{\frac{\text{Csc}[\theta]^2\,\Phi[]\;w^0}{r^2},\frac{\text{Csc}[\theta]^2\,\Phi[]\;w^1}{r^2},\frac{\text{Csc}[\theta]^2\,\Phi[]\;w^2}{r^2},\frac{\text{Csc}[\theta]^2\,\Phi[]\;w^3}{r^2}\right\},\{\text{ch}\},0\right]$$

Contract directly:

In[ ]:=
```
(scalar[] ToCTensor[w, {ch}] / (r[]^2 Sin[θ[]]^2))[μ] ω[-μ]
```

Out[ ]=

$$-\frac{2\,\text{Csc}[\theta]^2\,\Phi[]\;w^0}{r^2}+\frac{\text{Csc}[\theta]^2\,\Phi[]\;w^1}{r^2}-\frac{4\,\text{Csc}[\theta]^2\,\Phi[]\;w^2}{r^2}+\frac{3\,\text{Csc}[\theta]^2\,\Phi[]\;w^3}{r^2}$$

A (0,2) CTensor:

In[ ]:=
```
B = CTensor[{
    {1, 2, 3, 4},
    {2, 2, 5, 6},
    {3, 5, 3, 7},
    {4, 6, 7, 4}
  }, {-ch, -ch}]
```

Out[ ]=
```
CTensor[{{1, 2, 3, 4}, {2, 2, 5, 6}, {3, 5, 3, 7}, {4, 6, 7, 4}}, {-ch, -ch}, 0]
```

In[ ]:=
```
B[-μ, -ν]
```

Out[ ]=

$$\begin{pmatrix}1&2&3&4\\2&2&5&6\\3&5&3&7\\4&6&7&4\end{pmatrix}_{\mu\nu}$$

In[ ]:=
```
B[-μ, -ν] // ToBasisExpand
```

Out[ ]=

$$e_\mu{}^0\;e_\nu{}^0+2\,e_\mu{}^1\;e_\nu{}^0+3\,e_\mu{}^2\;e_\nu{}^0+4\,e_\mu{}^3\;e_\nu{}^0+2\,e_\mu{}^0\;e_\nu{}^1+$$
$$2\,e_\mu{}^1\;e_\nu{}^1+5\,e_\mu{}^2\;e_\nu{}^1+6\,e_\mu{}^3\;e_\nu{}^1+3\,e_\mu{}^0\;e_\nu{}^2+5\,e_\mu{}^1\;e_\nu{}^2+$$
$$3\,e_\mu{}^2\;e_\nu{}^2+7\,e_\mu{}^3\;e_\nu{}^2+4\,e_\mu{}^0\;e_\nu{}^3+6\,e_\mu{}^1\;e_\nu{}^3+7\,e_\mu{}^2\;e_\nu{}^3+4\,e_\mu{}^3\;e_\nu{}^3$$

```
In[ ]:= {B[-μ, -ν] v[μ],
          B[-μ, -ν] v[μ] // ToBasisExpand}
```

$$\left\{\begin{vmatrix}25\\41\\38\\39\end{vmatrix}_{\nu},\ 25\, e_{\nu}{}^{0}+41\, e_{\nu}{}^{1}+38\, e_{\nu}{}^{2}+39\, e_{\nu}{}^{3}\right\}$$

```
In[ ]:= {B[-μ, -ν] v[μ] v[ν],
          B[-μ, -ν] v[μ] w[ν],
          B[-μ, -ν] v[μ] w[ν] // ToBasisExpand,
          B[-μ, -ν] v[μ] w[ν] // ToBasisExpand // ContractBasis}
```

$$\left\{279,\ w^{\nu}\begin{vmatrix}25\\41\\38\\39\end{vmatrix}_{\nu},\ \left(25\, e_{\nu}{}^{0}+41\, e_{\nu}{}^{1}+38\, e_{\nu}{}^{2}+39\, e_{\nu}{}^{3}\right) w^{\nu},\ 25\, w^{0}+41\, w^{1}+38\, w^{2}+39\, w^{3}\right\}$$

This is a 1-form:

```
In[ ]:= 2 s[] B[-μ, -ν] v[μ] + 7 scalar[] ω[-ν] // ToBasisExpand
```

$$e_{\nu}{}^{2}\ (76\, s[]-28\, \Phi[])+e_{\nu}{}^{0}\ (50\, s[]-14\, \Phi[])+e_{\nu}{}^{1}\ (82\, s[]+7\, \Phi[])+e_{\nu}{}^{3}\ (78\, s[]+21\, \Phi[])$$

The Metric

To define a metric, we fist have to define a symmetric (0,2)-CTensor object:

```
In[ ]:= $Assumptions =
          mass > 0 && r[] > 2 mass && 0 < θ[] < 2 π && 0 < ϕ[] < 2 π && t[] ∈ Reals;
```

```
In[ ]:= g = CTensor[
          DiagonalMatrix[
           {-1 + 2 mass / r[], 1 / (1 - 2 mass / r[]), r[]^2, r[]^2 Sin[θ[]]^2}
          ], {-ch, -ch}]
```

$$\text{CTensor}\left[\left\{\left\{-1+\frac{2M}{r},0,0,0\right\},\left\{0,\frac{1}{1-\frac{2M}{r}},0,0\right\},\{0,0,r^2,0\},\{0,0,0,r^2\sin[\theta]^2\}\right\},\{-ch,-ch\},0\right]$$

In[ ]:=
```
{g[-μ, -ν],
 g[-μ, -ν] // ToBasisExpand}
```

Out[ ]=

$$\left\{\begin{pmatrix} -1+\frac{2M}{r} & 0 & 0 & 0 \\ 0 & \frac{1}{1-\frac{2M}{r}} & 0 & 0 \\ 0 & 0 & r^2 & 0 \\ 0 & 0 & 0 & r^2\,\text{Sin}[\theta]^2 \end{pmatrix}_{\mu\nu},\ \frac{e_\mu{}^1\ e_\nu{}^1}{1-\frac{2M}{r}} + e_\mu{}^0\ e_\nu{}^0\left(-1+\frac{2M}{r}\right) + e_\mu{}^2\ e_\nu{}^2\ r^2 + e_\mu{}^3\ e_\nu{}^3\ r^2\,\text{Sin}[\theta]^2\right\}$$

We declare $g_{\mu\nu}$ to be the metric on the chart ch using the SetCMetric function. We have to declare its signature

SignatureOfMetric → {n1,n2,n3}

n1 = number of (+1)

n2= number of (-1)

n3= number of (0) - degenerate metrics

In[ ]:=
```
SetCMetric[g, ch, SignatureOfMetric → {3, 1, 0}]
```

The covariant derivative, a Levi-Civita connection (compatible with g and torsion free) is defined by CovDOfMetric.

Notice that a connection in xCoba is an object with CCovD head. It includes CTensor information for Christoffel symbols and the metric itself. PDch is the partial derivative of the ch-chart.

We will do connection calculations later.

In[ ]:=
```
cd = CovDOfMetric[g]
```

Out[ ]=

$$\text{CCovD}\Big[\text{PDch},\ \text{CTensor}\Big[\Big\{\Big\{\Big\{0,-\frac{M}{2\,M\,r-r^2},0,0\Big\},\Big\{-\frac{M}{2\,M\,r-r^2},0,0,0\Big\},\{0,0,0,0\},\{0,0,0,0\}\Big\},$$
$$\Big\{\Big\{\frac{M\,(-2\,M+r)}{r^3},0,0,0\Big\},\Big\{0,\frac{M}{2\,M\,r-r^2},0,0\Big\},$$
$$\{0,0,2\,M-r,0\},\Big\{0,0,0,(2\,M-r)\,\text{Sin}[\theta]^2\Big\}\Big\},$$
$$\Big\{\{0,0,0,0\},\Big\{0,0,\frac{1}{r},0\Big\},\Big\{0,\frac{1}{r},0,0\Big\},\{0,0,0,-\text{Cos}[\theta]\,\text{Sin}[\theta]\}\Big\},$$
$$\Big\{\{0,0,0,0\},\Big\{0,0,0,\frac{1}{r}\Big\},\{0,0,0,\text{Cot}[\theta]\},\Big\{0,\frac{1}{r},\text{Cot}[\theta],0\Big\}\Big\}\Big\},\{\text{ch},-\text{ch},-\text{ch}\},0\Big],$$
$$\text{CTensor}\Big[\Big\{\Big\{-1+\frac{2\,M}{r},0,0,0\Big\},\Big\{0,\frac{1}{1-\frac{2\,M}{r}},0,0\Big\},\{0,0,r^2,0\},\Big\{0,0,0,r^2\,\text{Sin}[\theta]^2\Big\}\Big\},$$
$$\{-\text{ch},-\text{ch}\},0\Big]\Big]$$

An ϵ-Levi-Civita tensor is automatically defined:

In[ ]:=
```
epsilon[g]
```

Out[ ]=

```
CTensor[SparseArray[ Specified elements: 24
                     Dimensions: {4, 4, 4, 4} ], {-ch, -ch, -ch, -ch}, 0]
```

See its component structure:

```
In[ ]:= epsilon[g] // Normal // Simplify

Out[ ]= CTensor[{{{{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}},
    {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, Abs[Sin[θ]] r^2}, {0, 0, -Abs[Sin[θ]] r^2, 0}},
    {{0, 0, 0, 0}, {0, 0, 0, -Abs[Sin[θ]] r^2}, {0, 0, 0, 0}, {0, Abs[Sin[θ]] r^2, 0, 0}},
    {{0, 0, 0, 0}, {0, 0, Abs[Sin[θ]] r^2, 0}, {0, -Abs[Sin[θ]] r^2, 0, 0}, {0, 0, 0, 0}}},
   {{{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, -Abs[Sin[θ]] r^2}, {0, 0, Abs[Sin[θ]] r^2, 0}},
    {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}},
    {{0, 0, 0, Abs[Sin[θ]] r^2}, {0, 0, 0, 0}, {0, 0, 0, 0}, {-Abs[Sin[θ]] r^2, 0, 0, 0}},
    {{0, 0, -Abs[Sin[θ]] r^2, 0}, {0, 0, 0, 0}, {Abs[Sin[θ]] r^2, 0, 0, 0}, {0, 0, 0, 0}}},
   {{{0, 0, 0, 0}, {0, 0, 0, Abs[Sin[θ]] r^2}, {0, 0, 0, 0}, {0, -Abs[Sin[θ]] r^2, 0, 0}},
    {{0, 0, 0, -Abs[Sin[θ]] r^2}, {0, 0, 0, 0}, {0, 0, 0, 0}, {Abs[Sin[θ]] r^2, 0, 0, 0}},
    {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}},
    {{0, Abs[Sin[θ]] r^2, 0, 0}, {-Abs[Sin[θ]] r^2, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}},
   {{{0, 0, 0, 0}, {0, 0, -Abs[Sin[θ]] r^2, 0}, {0, Abs[Sin[θ]] r^2, 0, 0}, {0, 0, 0, 0}},
    {{0, 0, Abs[Sin[θ]] r^2, 0}, {0, 0, 0, 0}, {-Abs[Sin[θ]] r^2, 0, 0, 0}, {0, 0, 0, 0}},
    {{0, -Abs[Sin[θ]] r^2, 0, 0}, {Abs[Sin[θ]] r^2, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}},
    {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}}}, {-ch, -ch, -ch, -ch}, 0]
```

The determinant of the metric is in Determinant[g,ch]: It has weight 2, so we see the two blue tildes above its value.

```
In[ ]:= {Determinant[g, ch], Determinant[g, ch][]}

Out[ ]= {CTensor[-r^4 Sin[θ]^2, {}, 2 ch], -r^4 Sin[θ]^2}
```

The value of the determinant can be obtained directly using Mathematica's Det and ComponentArray to obtain a matrix:
(do not present at this point, show later with ComponentArray)

```
In[ ]:= Det[ComponentArray[g[{-μ, -ch}, {-ν, -ch}]]] // Simplify

Out[ ]= -r^4 Sin[θ]^2
```

We can raise/lower indices now:

In[ ]:=
```
{v[μ],
  v[-μ],
  v[μ] // ToBasisExpand,
  v[-μ] // ToBasisExpand,
  ω[-μ] v[μ],
  ω[μ] v[-μ] // Simplify,
  v[-μ] w[μ] // ToBasisExpand // ContractBasis // Simplify,
  B[-μ, -ν],
  B[μ, -ν],
  B[μ, ν] // Simplify,
  B[μ, ν] B[-μ, -ν],
  g[μ, ν],
  g[-μ, -ν] g[ν, ρ] // Simplify,
  g[-μ, -ν] v[ν] - v[-μ] // Simplify,
  g[-μ, -ν] w[ν] // ToBasisExpand // ContractBasis
}
```

Out[ ]=

$$\left\{\begin{pmatrix}2\\-1\\3\\4\end{pmatrix}^{\mu},\ \begin{pmatrix}-2+\frac{4M}{r}\\ \frac{r}{2M-r}\\ 3r^2\\ 4r^2\,\text{Sin}[\theta]^2\end{pmatrix}_{\mu},\ 2\,e_0{}^{\mu}-e_1{}^{\mu}+3\,e_2{}^{\mu}+4\,e_3{}^{\mu},\right.$$

$$e_{\mu}{}^{0}\left(-2+\frac{4M}{r}\right)+\frac{e_{\mu}{}^{1}\,r}{2M-r}+3\,e_{\mu}{}^{2}\,r^2+4\,e_{\mu}{}^{3}\,r^2\,\text{Sin}[\theta]^2,\ -5,\ -5,$$

$$\left(-2+\frac{4M}{r}\right)w^0+r\left(\frac{w^1}{2M-r}+3\,r\,w^2+4\,r\,\text{Sin}[\theta]^2\,w^3\right),\ \begin{pmatrix}1&2&3&4\\2&2&5&6\\3&5&3&7\\4&6&7&4\end{pmatrix}_{\mu\nu},\ \begin{pmatrix}\frac{r}{2M-r}&\frac{2r}{2M-r}&\frac{3r}{2M-r}&\frac{4r}{2M-r}\\ 2-\frac{4M}{r}&2-\frac{4M}{r}&5-\frac{10M}{r}&6-\frac{12M}{r}\\ \frac{3}{r^2}&\frac{5}{r^2}&\frac{3}{r^2}&\frac{7}{r^2}\\ \frac{4\,\text{Csc}[\theta]^2}{r^2}&\frac{6\,\text{Csc}[\theta]^2}{r^2}&\frac{7\,\text{Csc}[\theta]^2}{r^2}&\frac{4\,\text{Csc}[\theta]^2}{r^2}\end{pmatrix}^{\mu}{}_{\nu},$$

$$\begin{pmatrix}\frac{r^2}{(-2M+r)^2}&-2&\frac{3}{2Mr-r^2}&\frac{4\,\text{Csc}[\theta]^2}{2Mr-r^2}\\ -2&\frac{2(-2M+r)^2}{r^2}&\frac{5(-2M+r)}{r^3}&-\frac{6\,\text{Csc}[\theta]^2(2M-r)}{r^3}\\ \frac{3}{2Mr-r^2}&\frac{5(-2M+r)}{r^3}&\frac{3}{r^4}&\frac{7\,\text{Csc}[\theta]^2}{r^4}\\ \frac{4\,\text{Csc}[\theta]^2}{2Mr-r^2}&-\frac{6\,\text{Csc}[\theta]^2(2M-r)}{r^3}&\frac{7\,\text{Csc}[\theta]^2}{r^4}&\frac{4\,\text{Csc}[\theta]^4}{r^4}\end{pmatrix}^{\mu\nu},\ -8+\frac{9}{r^4}+\frac{98\,\text{Csc}[\theta]^2}{r^4}+\frac{16\,\text{Csc}[\theta]^4}{r^4}-\frac{72\,\text{Csc}[\theta]^2(2M-r)}{r^3}+$$

$$\frac{r^2}{(-2M+r)^2}+\frac{50(-2M+r)}{r^3}+\frac{4(-2M+r)^2}{r^2}+\frac{18}{2Mr-r^2}+\frac{32\,\text{Csc}[\theta]^2}{2Mr-r^2},\ \begin{pmatrix}\frac{r}{2M-r}&0&0&0\\0&1-\frac{2M}{r}&0&0\\0&0&\frac{1}{r^2}&0\\0&0&0&\frac{\text{Csc}[\theta]^2}{r^2}\end{pmatrix}^{\mu\nu},$$

$$\left.\begin{pmatrix}1&0&0&0\\0&1&0&0\\0&0&1&0\\0&0&0&1\end{pmatrix}^{\rho}{}_{\mu},\ 0,\ -e_{\mu}{}^{0}\,w^0+\frac{2M\,e_{\mu}{}^{0}\,w^0}{r}+\frac{e_{\mu}{}^{1}\,w^1}{1-\frac{2M}{r}}+e_{\mu}{}^{2}\,r^2\,w^2+e_{\mu}{}^{3}\,r^2\,\text{Sin}[\theta]^2\,w^3\right\}$$

Extract components of a tensor using ComponentArray for further manipulation: use Basis indices

In[ ]:=
```
{ComponentArray[v[{ μ,  ch}]] // MatrixForm,
 ComponentArray[v[{-μ, -ch}]] // MatrixForm,
 ComponentArray[w[{ μ,  ch}]] // MatrixForm,
 ComponentArray[u[{ μ,  ch}]] // MatrixForm,
 ComponentArray[B[{-μ, -ch}, {-ν, -ch}]] // MatrixForm,
 ComponentArray[B[{-μ, -ch}, { ν,  ch}]] // MatrixForm,
 ComponentArray[B[{-μ, -ch}, {-ν, -ch}] v[{ μ,  ch}] // ContractBasis] // MatrixForm,
 ComponentArray[
   B[{ μ,  ch}, {-ν, -ch}] v[{-μ, -ch}] // ContractBasis // Simplify] // MatrixForm
}
```

Out[ ]=

$$\left\{\begin{pmatrix}2\\-1\\3\\4\end{pmatrix},\begin{pmatrix}-2+\frac{4M}{r}\\ \frac{r}{2M-r}\\ 3r^2\\ 4r^2\,\mathrm{Sin}[\theta]^2\end{pmatrix},\begin{pmatrix}w^0\\w^1\\w^2\\w^3\end{pmatrix},\begin{pmatrix}\left(1-\frac{2M}{r}\right)w^0\\ \left(1-\frac{2M}{r}\right)w^1\\ \left(1-\frac{2M}{r}\right)w^2\\ \left(1-\frac{2M}{r}\right)w^3\end{pmatrix},\right.$$

$$\left.\begin{pmatrix}1&2&3&4\\2&2&5&6\\3&5&3&7\\4&6&7&4\end{pmatrix},\begin{pmatrix}\frac{r}{2M-r}&2-\frac{4M}{r}&\frac{3}{r^2}&\frac{4\,\mathrm{Csc}[\theta]^2}{r^2}\\ \frac{2r}{2M-r}&2-\frac{4M}{r}&\frac{5}{r^2}&\frac{6\,\mathrm{Csc}[\theta]^2}{r^2}\\ \frac{3r}{2M-r}&5-\frac{10M}{r}&\frac{3}{r^2}&\frac{7\,\mathrm{Csc}[\theta]^2}{r^2}\\ \frac{4r}{2M-r}&6-\frac{12M}{r}&\frac{7}{r^2}&\frac{4\,\mathrm{Csc}[\theta]^2}{r^2}\end{pmatrix},\begin{pmatrix}25\\41\\38\\39\end{pmatrix},\begin{pmatrix}25\\41\\38\\39\end{pmatrix}\right\}$$

Tensor products:

In[ ]:=
```
{ComponentArray[v[{ μ,  ch}] v[{ ν,  ch}]] // MatrixForm,
 ComponentArray[v[{ μ,  ch}] w[{ ν,  ch}]] // MatrixForm,
 ComponentArray[w[{ μ,  ch}] w[{ ν,  ch}]] // MatrixForm
}
```

Out[ ]=

$$\left\{\begin{pmatrix}4&-2&6&8\\-2&1&-3&-4\\6&-3&9&12\\8&-4&12&16\end{pmatrix},\begin{pmatrix}2\,w^0&2\,w^1&2\,w^2&2\,w^3\\-w^0&-w^1&-w^2&-w^3\\3\,w^0&3\,w^1&3\,w^2&3\,w^3\\4\,w^0&4\,w^1&4\,w^2&4\,w^3\end{pmatrix},\begin{pmatrix}{w^0}^2&w^0\,w^1&w^0\,w^2&w^0\,w^3\\w^0\,w^1&{w^1}^2&w^1\,w^2&w^1\,w^3\\w^0\,w^2&w^1\,w^2&{w^2}^2&w^2\,w^3\\w^0\,w^3&w^1\,w^3&w^2\,w^3&{w^3}^2\end{pmatrix}\right\}$$

In[ ]:=
```
Det[ComponentArray[g[{-μ, -ch}, {-ν, -ch}]]] // Simplify
```

Out[ ]=

$$-r^4\,\mathrm{Sin}[\theta]^2$$

Specific components, using component indices:

In[ ]:=
```
Print[
 "v^1  = ", v[{1, -ch}], "\n",
 "g22  = ", g[{2, -ch}, {2, -ch}], "\n",
 "g    = ", Table[g[{i, -ch}, {i, -ch}], {i, 0, 3}], "\n",
 "g^-1 = ", Table[g[{i,  ch}, {i,  ch}], {i, 0, 3}]
]
```

$v^1 = \frac{r}{2M - r}$

$g_{22} = r^2$

$g = \left\{-1 + \frac{2M}{r}, \frac{1}{1 - \frac{2M}{r}}, r^2, r^2 \text{Sin}[\theta]^2\right\}$

$g^{-1} = \left\{\frac{r}{2M - r}, 1 - \frac{2M}{r}, \frac{1}{r^2}, \frac{\text{Csc}[\theta]^2}{r^2}\right\}$

Use MakeRule to make substitutions:

In[ ]:=
```
uwvrule = MakeRule[{w[μ], v[μ]}];
dwvrule = MakeRule[{w[-μ], v[-μ]}];
{w[-μ] w[μ],
 w[-μ] w[μ] /. uwvrule, ,
 w[-μ] w[μ] /. dwvrule,
 w[-μ] w[μ] /. uwvrule /. dwvrule // Simplify
}
```

Out[ ]=
$\left\{w_\mu\ w^\mu,\ w_\mu \begin{vmatrix} 2 \\ -1 \\ 3 \\ 4 \end{vmatrix}^\mu,\ \text{Null},\ w^\mu \begin{vmatrix} -2+\frac{4M}{r} \\ \frac{r}{2M-r} \\ 3\ r^2 \\ 4\ r^2\ \text{Sin}[\theta]^2 \end{vmatrix}_\mu,\ -4 + \frac{8M}{r} - \frac{r}{2M - r} + 9 r^2 + 16 r^2 \text{Sin}[\theta]^2\right\}$

In[ ]:=
```
ctrule = MakeRule[{w[μ], CTensor[{scalar[], s[], r[]^2 Sin[θ[]], ℯ^t[]}, {ch}][μ]}];
{w[μ],
 w[μ]        /. ctrule // ToBasisExpand,
 w[μ] ω[-μ]  /. ctrule}
```

Out[ ]=
$\{w^\mu,\ e^t\ e_3{}^\mu + e_1{}^\mu\ s[] + e_0{}^\mu\ \Phi[] + e_2{}^\mu\ r^2\ \text{Sin}[\theta],\ 3\ e^t + s[] - 2\ \Phi[] - 4\ r^2\ \text{Sin}[\theta]\}$

In[ ]:=
```
ξ = CTensor[{1/Sqrt[1 - 2 mass/r[]], Sqrt[(1 - 2 mass/r[])], r[], r[] Cos[θ[]]  }, {-ch}];

F = CTensor[
  {
   {0          , 1 - 2 mass/r[], 0, Cos[θ[]]},
   {-1 + 2 mass/r[], 0          , 0,    0   },
   {0          , 0         , 0 ,    0   },
   {-Cos[θ[]] , 0          , 0 ,    0   }
  }, {ch, ch}];
{ComponentArray[ξ[{-μ, -ch}]] // MatrixForm,
 ComponentArray[F[{μ, ch}, {ν, ch}]] // MatrixForm}
```

Out[ ]=

$$\left\{\begin{pmatrix} \frac{1}{\sqrt{1-\frac{2M}{r}}} \\ \sqrt{1-\frac{2M}{r}} \\ r \\ \mathrm{Cos}[\theta]\, r \end{pmatrix}, \begin{pmatrix} 0 & 1-\frac{2M}{r} & 0 & \mathrm{Cos}[\theta] \\ -1+\frac{2M}{r} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ -\mathrm{Cos}[\theta] & 0 & 0 & 0 \end{pmatrix}\right\}$$

antisymmetric × symmetric = 0

In[ ]:=
```
{
 F[α, β] ξ[-α] ξ[-β]  // Simplify,
 F[μ, ν] B[-μ, -ν] // Simplify,
 epsilon[g][-μ, -ν, -ρ, -σ] B[μ, ν],
 epsilon[g][-μ, -ν, -ρ, -σ] F[μ, ν] // Simplify,
 epsilon[g][-μ, -ν, -ρ, -σ] F[μ, ν] F[ρ, σ] // Simplify,
 epsilon[g][-μ, -ν, -ρ, -σ] F[μ, ν] u[ρ] u[σ]
}
```

Out[ ]=

$$\left\{0, 0, 0, \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 2\ \mathrm{Abs}[\mathrm{Sin}[\theta]]\ \mathrm{Cos}[\theta]\ r^2 & 0 \\ 0 & -2\ \mathrm{Abs}[\mathrm{Sin}[\theta]]\ \mathrm{Cos}[\theta]\ r^2 & 0 & 2\ \mathrm{Abs}[\mathrm{Sin}[\theta]]\ r\ (-2\ M+r) \\ 0 & 0 & -2\ \mathrm{Abs}[\mathrm{Sin}[\theta]]\ r\ (-2\ M+r) & 0 \end{pmatrix}_{\rho\sigma}, 0, 0\right\}$$

Covariant Derivatives

In[ ]:=
```
{cd[-μ][v[μ]],
 cd[-μ]@g[-ν, -ρ],
 cd[-μ]@g[ ν,  ρ],
 cd[-μ]@w[μ],
 cd[-μ]@ξ[-ν] // Simplify, ,
 cd[-μ]@F[ν, ρ] // Simplify,
 cd[-μ]@F[μ, ν] // Simplify,
 cd[-μ] @ cd[-ν] @ v[ρ] // Simplify,
 cd[-μ] @ cd[-ν] @ v[μ] // Simplify
}
```

Out[ ]=

$$\Big\{3\,\mathrm{Cot}[\theta]-\frac{2}{r},\ 0,\ 0,\ \nabla_{\mu}w^{\mu},\ \begin{pmatrix} -\frac{M\,(-2\,M+r)^{3/2}}{r^{7/2}} & -\frac{2\,M}{\sqrt{r}\,(-2\,M+r)^{3/2}} & 0 & 0 \\ -\frac{M}{\sqrt{r}\,(-2\,M+r)^{3/2}} & \frac{2\,M}{\sqrt{r^3\,(-2\,M+r)}} & -1 & -\mathrm{Cos}[\theta] \\ 0 & 0 & \frac{(-2\,M+r)^{3/2}}{\sqrt{r}} & -\mathrm{Cos}[\theta]\,\mathrm{Cot}[\theta]\,r \\ 0 & 0 & -\mathrm{Csc}[\theta]\,r\,\mathrm{Cos}[\theta]\,r\,\mathrm{Sin}[\theta]+\frac{(-2\,M+r)^{3/2}\,\mathrm{Sin}[\theta]^2}{\sqrt{r}} & \end{pmatrix}_{\nu\mu},$$

$$\mathrm{Null},\ \begin{pmatrix} \begin{vmatrix}0\\0\\0\\0\end{vmatrix} & \begin{vmatrix}0\\ \frac{2\,M}{r^2}\\ 0\\ \mathrm{Cos}[\theta]\,(2\,M-r)\,\mathrm{Sin}[\theta]^2\end{vmatrix} & \begin{vmatrix}0\\0\\ \frac{-2\,M+r}{r^2}\\ -\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta]\end{vmatrix} & \begin{vmatrix}0\\ \frac{\mathrm{Cos}[\theta]\,(-M+r)}{r\,(-2\,M+r)}\\ \mathrm{Cos}[2\,\theta]\,\mathrm{Csc}[\theta]\\ \frac{-2\,M+r}{r^2}\end{vmatrix} \\ \begin{vmatrix}0\\ -\frac{2\,M}{r^2}\\ 0\\ -\mathrm{Cos}[\theta]\,(2\,M-r)\,\mathrm{Sin}[\theta]^2\end{vmatrix} & \begin{vmatrix}0\\0\\0\\0\end{vmatrix} & \begin{vmatrix}0\\0\\0\\0\end{vmatrix} & \begin{vmatrix}\frac{M\,\mathrm{Cos}[\theta]\,(-2\,M+r)}{r^3}\\0\\0\\0\end{vmatrix} \\ \begin{vmatrix}0\\0\\ \frac{2\,M-r}{r^2}\\ \mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta]\end{vmatrix} & \begin{vmatrix}0\\0\\0\\0\end{vmatrix} & \begin{vmatrix}0\\0\\0\\0\end{vmatrix} & \begin{vmatrix}0\\0\\0\\0\end{vmatrix} \\ \begin{vmatrix}0\\ \frac{\mathrm{Cos}[\theta]\,(M-r)}{r\,(-2\,M+r)}\\ -\mathrm{Cos}[2\,\theta]\,\mathrm{Csc}[\theta]\\ \frac{2\,M-r}{r^2}\end{vmatrix} & \begin{vmatrix}\frac{M\,\mathrm{Cos}[\theta]\,(2\,M-r)}{r^3}\\0\\0\\0\end{vmatrix} & \begin{vmatrix}0\\0\\0\\0\end{vmatrix} & \begin{vmatrix}0\\0\\0\\0\end{vmatrix} \end{pmatrix}^{\nu\rho}{}_{\mu},$$

$$\begin{vmatrix}\frac{2\,(M-r)}{r^2}\\0\\0\\0\end{vmatrix}_{\nu},\ \begin{pmatrix} \begin{vmatrix}0\\ -\frac{2\,M\,(M-r)}{r^2\,(-2\,M+r)^2}\\ 0\\ 0\end{vmatrix} & \begin{vmatrix}\frac{2\,M^2}{r^2\,(-2\,M+r)^2}\\ \frac{4\,M}{(2\,M-r)\,r^2}\\ 0\\ 0\end{vmatrix} & \begin{vmatrix}-\frac{3\,M}{r}\\ 0\\ \frac{2\,M}{r}\\ 0\end{vmatrix} & \begin{vmatrix}-\frac{4\,M\,\mathrm{Sin}[\theta]^2}{r}\\ 0\\ 0\\ \frac{2\,M\,\mathrm{Sin}[\theta]^2}{r}\end{vmatrix} \\ \begin{vmatrix}-\frac{2\,M^2}{r^4}\\ \frac{4\,M\,(2\,M-r)}{r^4}\\ 0\\ 0\end{vmatrix} & \begin{vmatrix}0\\ \frac{2\,M\,(M-r)}{r^2\,(-2\,M+r)^2}\\ 0\\ 0\end{vmatrix} & \begin{vmatrix}0\\ -\frac{3\,M}{r}\\ 1-\frac{M}{r}\\ 0\end{vmatrix} & \begin{vmatrix}0\\ -\frac{4\,M\,\mathrm{Sin}[\theta]^2}{r}\\ 0\\ 0\end{vmatrix} \\ \begin{vmatrix}-\frac{3\,M\,(-2\,M+r)}{r^4}\\ 0\\ \frac{2\,M\,(-2\,M+r)}{r^4}\\ 0\end{vmatrix} & \begin{vmatrix}0\\ -\frac{3\,M}{(2\,M-r)\,r^2}\\ \frac{-M+r}{r^2\,(-2\,M+r)}\\ 0\end{vmatrix} & \begin{vmatrix}0\\ \frac{1}{r^2}\\ 0\\ 0\end{vmatrix} & \begin{vmatrix}0\\ 0\\ \frac{8\,M\,\mathrm{Sin}[\theta]^2}{r}\\ -\frac{3\,(M+\mathrm{Cos}[2\,\theta]\,(-M+r))}{r}\end{vmatrix} \\ \begin{vmatrix}-\frac{4\,M\,(-2\,M+r)}{r^4}\\ 0\\ 0\\ \frac{2\,M\,(-2\,M+r)}{r^4}\end{vmatrix} & \begin{vmatrix}0\\ -\frac{4\,M}{(2\,M-r)\,r^2}\\ 0\\ \frac{-M+r}{r^2\,(-2\,M+r)}\end{vmatrix} & \begin{vmatrix}0\\ 0\\ -\frac{8\,M}{r}\\ -3-3\,\mathrm{Cot}[\theta]^2+\frac{6\,M}{r}\end{vmatrix} & \begin{vmatrix}0\\ \frac{1}{r^2}\\ -3\,\mathrm{Csc}[\theta]^2\\ 0\end{vmatrix} \end{pmatrix}^{\rho}{}_{\nu\mu},\ \begin{vmatrix}0\\ \frac{2}{r^2}\\ -3\,\mathrm{Csc}[\theta]^2\\ 0\end{vmatrix}_{\nu}\Big\}$$

In[∘]:=
```
Christoffel[cd, PDch][μ, -ν, -ρ]
```

Out[∘]=

$$\left(\begin{array}{cccc}
\begin{pmatrix}0\\ -\frac{M}{2\,M\,r-r^2}\\ 0\\ 0\end{pmatrix} & \begin{pmatrix}-\frac{M}{2\,M\,r-r^2}\\ 0\\ 0\\ 0\end{pmatrix} & \begin{pmatrix}0\\ 0\\ 0\\ 0\end{pmatrix} & \begin{pmatrix}0\\ 0\\ 0\\ 0\end{pmatrix}\\
\begin{pmatrix}\frac{M\,(-2\,M+r)}{r^3}\\ 0\\ 0\\ 0\end{pmatrix} & \begin{pmatrix}0\\ \frac{M}{2\,M\,r-r^2}\\ 0\\ 0\end{pmatrix} & \begin{pmatrix}0\\ 0\\ 2\,M-r\\ 0\end{pmatrix} & \begin{pmatrix}0\\ 0\\ 0\\ (2\,M-r)\,\mathrm{Sin}[\theta]^2\end{pmatrix}\\
\begin{pmatrix}0\\ 0\\ 0\\ 0\end{pmatrix} & \begin{pmatrix}0\\ 0\\ \frac{1}{r}\\ 0\end{pmatrix} & \begin{pmatrix}0\\ \frac{1}{r}\\ 0\\ 0\end{pmatrix} & \begin{pmatrix}0\\ 0\\ 0\\ -\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\end{pmatrix}\\
\begin{pmatrix}0\\ 0\\ 0\\ 0\end{pmatrix} & \begin{pmatrix}0\\ 0\\ 0\\ \frac{1}{r}\end{pmatrix} & \begin{pmatrix}0\\ 0\\ 0\\ \mathrm{Cot}[\theta]\end{pmatrix} & \begin{pmatrix}0\\ \frac{1}{r}\\ \mathrm{Cot}[\theta]\\ 0\end{pmatrix}
\end{array}\right)^{\mu}{}_{\nu\rho}$$

In[∘]:=
```
Christoffel[cd, PDch][-μ, -ν, -ρ]
```

Out[∘]=

$$\left(\begin{array}{cccc}
\begin{pmatrix}0\\ -\frac{M}{r^2}\\ 0\\ 0\end{pmatrix} & \begin{pmatrix}-\frac{M}{r^2}\\ 0\\ 0\\ 0\end{pmatrix} & \begin{pmatrix}0\\ 0\\ 0\\ 0\end{pmatrix} & \begin{pmatrix}0\\ 0\\ 0\\ 0\end{pmatrix}\\
\begin{pmatrix}\frac{M}{r^2}\\ 0\\ 0\\ 0\end{pmatrix} & \begin{pmatrix}0\\ -\frac{M}{(-2\,M+r)^2}\\ 0\\ 0\end{pmatrix} & \begin{pmatrix}0\\ 0\\ -r\\ 0\end{pmatrix} & \begin{pmatrix}0\\ 0\\ 0\\ -r\,\mathrm{Sin}[\theta]^2\end{pmatrix}\\
\begin{pmatrix}0\\ 0\\ 0\\ 0\end{pmatrix} & \begin{pmatrix}0\\ 0\\ r\\ 0\end{pmatrix} & \begin{pmatrix}0\\ r\\ 0\\ 0\end{pmatrix} & \begin{pmatrix}0\\ 0\\ 0\\ -\mathrm{Cos}[\theta]\,r^2\,\mathrm{Sin}[\theta]\end{pmatrix}\\
\begin{pmatrix}0\\ 0\\ 0\\ 0\end{pmatrix} & \begin{pmatrix}0\\ 0\\ 0\\ r\,\mathrm{Sin}[\theta]^2\end{pmatrix} & \begin{pmatrix}0\\ 0\\ 0\\ \mathrm{Cos}[\theta]\,r^2\,\mathrm{Sin}[\theta]\end{pmatrix} & \begin{pmatrix}0\\ r\,\mathrm{Sin}[\theta]^2\\ \mathrm{Cos}[\theta]\,r^2\,\mathrm{Sin}[\theta]\\ 0\end{pmatrix}
\end{array}\right)_{\mu\nu\rho}$$

In[∘]:=
```
Christoffel[cd, PDch][-μ, -ν, -ρ] v[ν] v[ρ] // Simplify
```

Out[∘]=

$$\begin{pmatrix}\frac{4\,M}{r^2}\\ \frac{4\,M}{r^2}-9\,r-\frac{M}{(-2\,M+r)^2}-16\,r\,\mathrm{Sin}[\theta]^2\\ -2\,r\,(3+4\,r\,\mathrm{Sin}[2\,\theta])\\ 8\,r\,(3\,\mathrm{Cos}[\theta]\,r-\mathrm{Sin}[\theta])\,\mathrm{Sin}[\theta]\end{pmatrix}_{\mu}$$

In[ ]:=
```
ComponentArray[Christoffel[cd, PDch][{μ, ch}, {-ν, -ch}, {-ρ, -ch}]] // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} \begin{pmatrix} 0 \\ -\frac{M}{2 M r-r^2} \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} -\frac{M}{2 M r-r^2} \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} \\ \begin{pmatrix} \frac{M(-2 M+r)}{r^3} \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ \frac{M}{2 M r-r^2} \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 2 M-r \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ (2 M-r)\,\mathrm{Sin}[\theta]^2 \end{pmatrix} \\ \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ \frac{1}{r} \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ \frac{1}{r} \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ -\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta] \end{pmatrix} \\ \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ \frac{1}{r} \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ \mathrm{Cot}[\theta] \end{pmatrix} & \begin{pmatrix} 0 \\ \frac{1}{r} \\ \mathrm{Cot}[\theta] \\ 0 \end{pmatrix} \end{pmatrix}$$

Curvature

In[ ]:=
```
Riemann[cd][-μ, -ν, -ρ, σ]
```

Out[ ]=

$$\begin{pmatrix} \begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} & \begin{pmatrix} 0&\frac{2 M (2 M-r)}{r^4}&0&0\\ \frac{2 M}{(2 M-r) r^2}&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} & \begin{pmatrix} 0&0&\frac{M (-2 M+r)}{r^4}&0\\0&0&0&0\\ \frac{M}{r}&0&0&0\\0&0&0&0 \end{pmatrix} & \begin{pmatrix} 0&0&0&\frac{M (-2 M+r)}{r^4}\\0&0&0&0\\0&0&0&0\\ \frac{M\,\mathrm{Sin}[\theta]^2}{r}&0&0&0 \end{pmatrix} \\ \begin{pmatrix} 0&\frac{2 M (-2 M+r)}{r^4}&0&0\\ -\frac{2 M}{(2 M-r) r^2}&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} & \begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} & \begin{pmatrix} 0&0&0&0\\0&0&\frac{M}{(2 M-r) r^2}&0\\0&\frac{M}{r}&0&0\\0&0&0&0 \end{pmatrix} & \begin{pmatrix} 0&0&0&0\\0&0&0&\frac{M}{(2 M-r) r^2}\\0&0&0&0\\0&\frac{M\,\mathrm{Sin}[\theta]^2}{r}&0&0 \end{pmatrix} \\ \begin{pmatrix} 0&0&\frac{M (2 M-r)}{r^4}&0\\0&0&0&0\\ -\frac{M}{r}&0&0&0\\0&0&0&0 \end{pmatrix} & \begin{pmatrix} 0&0&0&0\\0&0&-\frac{M}{(2 M-r) r^2}&0\\0&-\frac{M}{r}&0&0\\0&0&0&0 \end{pmatrix} & \begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} & \begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&\frac{2 M}{r}\\0&0&-\frac{2 M\,\mathrm{Sin}[\theta]^2}{r}&0 \end{pmatrix} \\ \begin{pmatrix} 0&0&0&\frac{M (2 M-r)}{r^4}\\0&0&0&0\\0&0&0&0\\ -\frac{M\,\mathrm{Sin}[\theta]^2}{r}&0&0&0 \end{pmatrix} & \begin{pmatrix} 0&0&0&0\\0&0&0&-\frac{M}{(2 M-r) r^2}\\0&0&0&0\\0&-\frac{M\,\mathrm{Sin}[\theta]^2}{r}&0&0 \end{pmatrix} & \begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&-\frac{2 M}{r}\\0&0&\frac{2 M\,\mathrm{Sin}[\theta]^2}{r}&0 \end{pmatrix} & \begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} \end{pmatrix}_{\mu\nu\rho}{}^{\sigma}$$

In[ ]:=
```
Ricci[cd][-μ, -ν]
```
Out[ ]=
```
0
```

In[ ]:=
```
RicciScalar[cd][]
```
Out[ ]=
```
0
```

In[ ]:=
```
Torsion[cd][α, -β, -γ]
```
Out[ ]=
```
0
```

In[ ]:=
```
Einstein[cd][-μ, -ν]
```
Out[ ]=
```
0
```

In[ ]:=
```
Weyl[cd][-μ, -ν, -ρ, -λ]
```
Out[ ]=

$$\begin{pmatrix}
\begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&-\frac{2M}{r^3}&0&0\\ \frac{2M}{r^3}&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&\frac{M(-2M+r)}{r^2}&0\\0&0&0&0\\ \frac{M(2M-r)}{r^2}&0&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&\frac{M(-2M+r)\,\mathrm{Sin}[\theta]^2}{r^2}\\0&0&0&0\\0&0&0&0\\ \frac{M(2M-r)\,\mathrm{Sin}[\theta]^2}{r^2}&0&0&0 \end{pmatrix} \\
\begin{pmatrix} 0&\frac{2M}{r^3}&0&0\\ -\frac{2M}{r^3}&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\0&0&\frac{M}{2M-r}&0\\0&-\frac{M}{2M-r}&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\0&0&0&\frac{M\,\mathrm{Sin}[\theta]^2}{2M-r}\\0&0&0&0\\0&-\frac{M\,\mathrm{Sin}[\theta]^2}{2M-r}&0&0 \end{pmatrix} \\
\begin{pmatrix} 0&0&\frac{M(2M-r)}{r^2}&0\\0&0&0&0\\ \frac{M(-2M+r)}{r^2}&0&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\0&0&-\frac{M}{2M-r}&0\\0&\frac{M}{2M-r}&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&2\,M\,r\,\mathrm{Sin}[\theta]^2\\0&0&-2\,M\,r\,\mathrm{Sin}[\theta]^2&0 \end{pmatrix} \\
\begin{pmatrix} 0&0&0&\frac{M(2M-r)\,\mathrm{Sin}[\theta]^2}{r^2}\\0&0&0&0\\0&0&0&0\\ -\frac{M(2M-r)\,\mathrm{Sin}[\theta]^2}{r^2}&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\0&0&0&-\frac{M\,\mathrm{Sin}[\theta]^2}{2M-r}\\0&0&0&0\\0&\frac{M\,\mathrm{Sin}[\theta]^2}{2M-r}&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&-2\,M\,r\,\mathrm{Sin}[\theta]^2\\0&0&2\,M\,r\,\mathrm{Sin}[\theta]^2&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix}
\end{pmatrix}_{\mu\nu\rho\lambda}$$

In[ ]:=
```
Riemann[cd][-μ, -ν, -ρ, -λ] == Weyl[cd][-μ, -ν, -ρ, -λ]
```
Out[ ]=
```
True
```

In[ ]:=
```
Kretschmann[cd][]
```
Out[ ]=

$$\frac{48\,M^2}{r^6}$$

Commutator of derivatives:

$[\nabla_\beta, \nabla_\alpha] v^\gamma = R_{\alpha\beta\delta}{}^{\gamma} v^\delta$

In[ ]:=
```
Print[
 "[∇β , ∇α] vγ = ", comm = cd[-β]@cd[-α]@v[γ] - cd[-α]@cd[-β]@v[γ] // Simplify, "\n",
 "    Rαβδγ vδ = ", riem = Riemann[cd][-α, -β, -δ, γ] v[δ], "\n",
 "[∇β , ∇α] vγ = Rαβδγ vδ    is ", riem == comm // Simplify
]
```

$$[\nabla_\beta , \nabla_\alpha]\, v^\gamma = \begin{pmatrix}
\begin{pmatrix} 0 \\ -\frac{2 M}{(2 M-r) r^2} \\ \frac{3 M}{r} \\ \frac{4 M \sin[\theta]^2}{r} \end{pmatrix} & \begin{pmatrix} \frac{2 M}{(2 M-r) r^2} \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} -\frac{3 M}{r} \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} -\frac{4 M \sin[\theta]^2}{r} \\ 0 \\ 0 \\ 0 \end{pmatrix} \\
\begin{pmatrix} 0 \\ \frac{4 M (2 M-r)}{r^4} \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} \frac{4 M (-2 M+r)}{r^4} \\ 0 \\ \frac{3 M}{r} \\ \frac{4 M \sin[\theta]^2}{r} \end{pmatrix} & \begin{pmatrix} 0 \\ -\frac{3 M}{r} \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ -\frac{4 M \sin[\theta]^2}{r} \\ 0 \\ 0 \end{pmatrix} \\
\begin{pmatrix} 0 \\ 0 \\ \frac{2 M (-2 M+r)}{r^4} \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ \frac{M}{r^2 (-2 M+r)} \\ 0 \end{pmatrix} & \begin{pmatrix} -\frac{2 M (-2 M+r)}{r^4} \\ \frac{M}{(2 M-r) r^2} \\ 0 \\ -\frac{8 M \sin[\theta]^2}{r} \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ \frac{8 M \sin[\theta]^2}{r} \\ 0 \end{pmatrix} \\
\begin{pmatrix} 0 \\ 0 \\ 0 \\ \frac{2 M (-2 M+r)}{r^4} \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ \frac{M}{r^2 (-2 M+r)} \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ \frac{6 M}{r} \end{pmatrix} & \begin{pmatrix} -\frac{2 M (-2 M+r)}{r^4} \\ \frac{M}{(2 M-r) r^2} \\ -\frac{6 M}{r} \\ 0 \end{pmatrix}
\end{pmatrix}^{\gamma}{}_{\alpha\beta}$$

$$R_{\alpha\beta\delta}{}^{\gamma}\, v^\delta = \begin{pmatrix}
\begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} -\frac{2 M}{(2 M-r) r^2} \\ \frac{4 M (2 M-r)}{r^4} \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} \frac{3 M}{r} \\ 0 \\ \frac{2 M (-2 M+r)}{r^4} \\ 0 \end{pmatrix} & \begin{pmatrix} \frac{4 M \sin[\theta]^2}{r} \\ 0 \\ 0 \\ \frac{2 M (-2 M+r)}{r^4} \end{pmatrix} \\
\begin{pmatrix} \frac{2 M}{(2 M-r) r^2} \\ \frac{4 M (-2 M+r)}{r^4} \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ \frac{3 M}{r} \\ -\frac{M}{(2 M-r) r^2} \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ \frac{4 M \sin[\theta]^2}{r} \\ 0 \\ -\frac{M}{(2 M-r) r^2} \end{pmatrix} \\
\begin{pmatrix} -\frac{3 M}{r} \\ 0 \\ \frac{2 M (2 M-r)}{r^4} \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ -\frac{3 M}{r} \\ \frac{M}{(2 M-r) r^2} \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ -\frac{8 M \sin[\theta]^2}{r} \\ \frac{6 M}{r} \end{pmatrix} \\
\begin{pmatrix} -\frac{4 M \sin[\theta]^2}{r} \\ 0 \\ 0 \\ \frac{2 M (2 M-r)}{r^4} \end{pmatrix} & \begin{pmatrix} 0 \\ -\frac{4 M \sin[\theta]^2}{r} \\ 0 \\ \frac{M}{(2 M-r) r^2} \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ \frac{8 M \sin[\theta]^2}{r} \\ -\frac{6 M}{r} \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix}
\end{pmatrix}_{\alpha\beta}{}^{\gamma}$$

$[\nabla_\beta , \nabla_\alpha] v^\gamma = R_{\alpha\beta\delta}{}^{\gamma} v^\delta$ is True

Basis

Coordinate basis:

In[ ]:=
```
Table[Basis[{i, -ch}, μ], {i, 0, dimM1}]
```
Out[ ]=

$\{e_0{}^\mu, e_1{}^\mu, e_2{}^\mu, e_3{}^\mu\}$

In[ ]:=
```
Table[Basis[{i, ch}, -μ], {i, 0, dimM1}]
```
Out[ ]=

$\{e_\mu{}^0, e_\mu{}^1, e_\mu{}^2, e_\mu{}^3\}$

In[ ]:=
```
BasisArray[-ch][μ]
```
Out[ ]=

$\{e_0{}^\mu, e_1{}^\mu, e_2{}^\mu, e_3{}^\mu\}$

In[ ]:=
```
BasisArray[-ch, -ch][μ, ν] // MatrixForm
```
Out[ ]//MatrixForm=

$$\begin{pmatrix} e_0{}^\mu e_0{}^\nu & e_0{}^\mu e_1{}^\nu & e_0{}^\mu e_2{}^\nu & e_0{}^\mu e_3{}^\nu \\ e_0{}^\nu e_1{}^\mu & e_1{}^\mu e_1{}^\nu & e_1{}^\mu e_2{}^\nu & e_1{}^\mu e_3{}^\nu \\ e_0{}^\nu e_2{}^\mu & e_1{}^\nu e_2{}^\mu & e_2{}^\mu e_2{}^\nu & e_2{}^\mu e_3{}^\nu \\ e_0{}^\nu e_3{}^\mu & e_1{}^\nu e_3{}^\mu & e_2{}^\nu e_3{}^\mu & e_3{}^\mu e_3{}^\nu \end{pmatrix}$$

Check duality of base: $e_\alpha{}^\mu e_\mu{}^\beta = \delta_\alpha{}^\beta$

In[ ]:=
```
Table[
  Basis[{i, -ch}, μ] Basis[{j, ch}, -μ]  // ContractBasis,    {i, 0, dimM1}, {j, 0, dimM1}
 ]                                          // MatrixForm
```
Out[ ]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

The components of tensors are the contractions with respective basis vectors:

$\xi_\alpha = e_\alpha{}^\sigma \xi_\sigma = \xi(e_\alpha)$, $\quad B_{\alpha\beta} = e_\alpha{}^\mu e_\beta{}^\nu B_{\mu\nu} = B(e_\alpha, e_\beta)$, $\quad v^\alpha = e_\mu{}^\alpha v^\mu = v(e^\alpha)$, ...

In[ ]:=
```
{ξ[{1, -ch}], ξ[-μ] Basis[{1, -ch}, μ] // ContractBasis,
 B[{1, -ch}, {2, -ch}], B[-μ, -ν] Basis[{1, -ch}, μ] Basis[{2, -ch}, ν] // ContractBasis,
 v[{1, ch}], v[μ] Basis[{1, ch}, -μ] // ContractBasis}
```

Out[ ]=

$$\left\{\sqrt{1-\frac{2\,M}{r}},\ \sqrt{1-\frac{2\,M}{r}},\ 5,\ 5,\ -1,\ -1\right\}$$

$$\partial_\mu \cdot \partial_\nu = g_{\mu\nu} \equiv g(\partial_\mu, \partial_\nu)$$

In[ ]:=
```
Table[
  g[-μ, -ν] Basis[{i, -ch}, μ] Basis[{j, -ch}, ν] // ContractBasis,
  {i, 0, dimM1}, {j, 0, dimM1}
 ] // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} -1+\frac{2\,M}{r} & 0 & 0 & 0 \\ 0 & \frac{1}{1-\frac{2\,M}{r}} & 0 & 0 \\ 0 & 0 & r^2 & 0 \\ 0 & 0 & 0 & r^2\,\mathrm{Sin}[\theta]^2 \end{pmatrix}$$

$\nabla_\mu e_1{}^\nu$, $\nabla_\mu e_\nu{}^2$, $\nabla_\mu\left(e_\nu{}^2\, e_\rho{}^0\right)$, $\nabla_\mu \nabla_\nu e_\rho{}^0$, $\nabla_\mu \nabla_\nu e_1{}^\mu$

In[ ]:=
```
{cd[-μ]@Basis[{1, -ch}, ν],
 cd[-μ]@Basis[{2, ch}, -ν],
 cd[-μ]@(Basis[{2, ch}, -ν] Basis[{0, ch}, -ρ]),
 cd[-μ]@cd[-ν]@ Basis[{0, ch}, -ρ],
 cd[-μ]@cd[-ν]@ Basis[{1, -ch}, μ] // Simplify
}
```

Out[ ]=

$$\left\{ \begin{pmatrix} -\frac{M}{2 M r-r^2} & 0 & 0 & 0 \\ 0 & \frac{M}{2 M r-r^2} & 0 & 0 \\ 0 & 0 & \frac{1}{r} & 0 \\ 0 & 0 & 0 & \frac{1}{r} \end{pmatrix}^{\nu}{}_{\mu},\ \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & -\frac{1}{r} & 0 \\ 0 & -\frac{1}{r} & 0 & 0 \\ 0 & 0 & 0 & \text{Cos}[\theta]\ \text{Sin}[\theta] \end{pmatrix}_{\mu\nu},\right.$$

$$e_{\rho}{}^{0} \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & -\frac{1}{r} & 0 \\ 0 & -\frac{1}{r} & 0 & 0 \\ 0 & 0 & 0 & \text{Cos}[\theta]\ \text{Sin}[\theta] \end{pmatrix}_{\mu\nu} + e_{\nu}{}^{2} \begin{pmatrix} 0 & \frac{M}{2 M r-r^2} & 0 & 0 \\ \frac{M}{2 M r-r^2} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}_{\mu\rho},$$

$$\begin{pmatrix} \begin{pmatrix} \frac{2 M^2}{r^4} \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ -\frac{2 M (M-r)}{r^2 (-2 M+r)^2} \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ -\frac{M}{r} \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ -\frac{M\ \text{Sin}[\theta]^2}{r} \end{pmatrix} \\ \begin{pmatrix} 0 \\ -\frac{2 M (M-r)}{r^2 (-2 M+r)^2} \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} \frac{2 M^2}{r^2 (-2 M+r)^2} \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} \\ \begin{pmatrix} 0 \\ 0 \\ -\frac{M}{r} \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} \\ \begin{pmatrix} 0 \\ 0 \\ 0 \\ -\frac{M\ \text{Sin}[\theta]^2}{r} \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} \end{pmatrix}_{\nu\rho\mu},\ \left. \begin{pmatrix} 0 \\ -\frac{2}{r^2} \\ 0 \\ 0 \end{pmatrix}_{\nu} \right\}$$

Acknowledgements

This notebook has been programmed by Konstantinos Anagnostopoulos, Physics Department, National Technical University of Athens, Greece, while he was an instructor of the 4th year undergraduate course "General Relativity and Cosmology". It was created for fun, but it may turn out to be useful to everyone studying the General Theory of Relativity for the first time.

Email: konstant@mail.ntua.gr

Web: http://physics.ntua.gr/konstant